/*
 * Licensed to the Apache Software Foundation (ASF) under one or more
 * contributor license agreements.  See the NOTICE file distributed with
 * this work for additional information regarding copyright ownership.
 * The ASF licenses this file to You under the Apache License, Version 2.0
 * (the "License"); you may not use this file except in compliance with
 * the License.  You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

grammar DALStatement;

import DMLStatement;

explain
    : DESCRIBE tableName
    | ANALYZE tableName
    ;

fromSchema
    : (FROM | IN) (owner DOT_)? schemaName
    ;

fromTable
    : (FROM | IN) tableName
    ;

showLike
    : LIKE stringLiterals
    ;

showWhereClause
    : WHERE expr
    ;

showFilter
    : showLike | showWhereClause
    ;

showColumns
    : SHOW EXTENDED? FULL? COLUMNS fromTable fromSchema? showFilter?
    ;

showFunctionCode
    : SHOW FUNCTION CODE functionName
    ;

showFunctions
    : SHOW FUNCTIONS
    ;

showTables
    : SHOW EXTENDED? FULL? TABLES fromSchema? showFilter?
    ;

showSession
    : SHOW SESSION
    ;

showCatalogs
    : SHOW CATALOGS
    ;

showSchemas
    : SHOW SCHEMAS (FROM databaseName)?
    ;

showCreateTable
    : SHOW CREATE TABLE tableName
    ;

show
    : showSession
    | showColumns
    | showFunctionCode
    | showFunctions
    | showCatalogs
    | showSchemas
    | showTables
    | showCreateTable
    ;
